

## MAX3286 I/O Model

SPICE I/O Macromodels aid in understanding signal integrity issues in electronic systems. Most of Maxim's High Frequency/Fiber Communication ICs utilize input and output (I/O) circuits with Current Mode Logic (CML), Positive Emitter Coupled Logic (PECL), and Low Voltage Differential Signal (LVDS) formats to transfer data into and out of an IC. These models are based on simplified circuit expressions that may include replacement of active circuit elements with ideal controlled voltage and current sources. As such, simulation with macromodels should be treated as 'typical' performance and not relied upon as final proof-of-design. Use of macromodel descriptions is not a substitute for worst-case design analysis, nor for testing real circuits over temperature, supply, and other operating limits.

The output format is provided as ASCII text netlists suitable for generic SPICE. This format is compatible with most versions of SPICE such as PSPICE and HSPICE. Additional information is found in HFAN 6.1 *Input/Output Models for Maxim Fiber Components*.

To extract the circuit netlists using the Adobe Acrobat Reader follow these instructions. Select the "Text Select Tool" by clicking the left mouse button on this icon of the menu bar (a capital T with a small dashed box to the lower right). Highlight the desired netlist text with the cursor. Use the copy command from the edit menu to capture the selected lines. Then paste the selected lines into a text file editor and save the file with an extension compatible with the simulator.

Version B, Mar 7, 2003  
Version B1, May 23, 2003

## Input Model MAX3286 LAN Laser Driver




Figure 2. Simplified input package model and input circuitry for the MAX3286.

## Notes:

The schematic on the previous page represents the input stage of the Maxim MAX3286 LAN Laser Driver. The output stage is not modeled since the electrical model is not considered an adequate representation for the analysis of the resultant optical signal. The MAX3286 uses the same die as the MAX3296, however the MAX3296 has a faster output drive. Thus there is no difference in the input model.

The netlist is in SPICE 2g6 format. Since nodes in SPICE 2g6 can only be numbers, the output signals are 2101 and 2102 and the input signals to the die are 1001 and 1002. Comments in the netlist identify the correspondence between the signal names and the node numbers. The netlists are in SPICE 2g6 format and are compatible with PSPICE and HSPICE. It has been simulated on a generic SPICE simulator and PSPICE.

**The Input Stage:** The input stage is comprised of two subcircuits INPKG and DRV\_IN.

**The PAD and ESD Model:** The pad and ESD model have been simplified somewhat and the simplifications have little effect on the signals.

**Text File Format:** This model is shipped in “pdf” format. Models and netlists can be copied to text format in the Acrobat Reader by holding the left mouse button on the “Text Select Tool.” Then the user can “select” what netlist and/or subcircuit with the mouse. Then use the copy command from the “edit” menu to capture the selected lines. These lines can then be “pasted” into the user’s text file.

## Circuit Netlist – Input circuit

INPUT - MAX3286 INPUT CIRCUIT

\*

\* THIS IS THE INPUT OF THE MAX3286

\* NOTE AS PER TYPICAL APPLICATION CIRCUIT A 115 OHM RESISTOR

\* HAS BEEN PLACED ACCROSS THE INPUTS TO THE DEVICE

\*

.OPTIONS ACCT NOMOD NOPAGE LIMPTS=10000 RELTOL=.001

.WIDTH OUT=80

.TEMP 80

\* TYPICAL DIE TEMP =  $25C + 2.2W*(26C/W) = 79C$

\*

\* INPUT SIGNALS +IN = 2101, -IN = 2102

.OP

.TRAN 5PS 2.6NS

VCC 101 0 DC 3.3

\* The power supply is 3.3Volts.

\*

VINA 2 0 PULSE (3.3 2.5 0.2N 0.100N 0.100N 0.8200N 1.880N)

VINB 3 0 PULSE (2.5 3.3 0.2N 0.100N 0.100N 0.820N 1.880N)

\*CS1 2 2a 0.01U IC=-.0

\*CS2 3 3a 0.01U IC=+.0

RLOAD1 2 2101 50

CLOAD1 101 2101 0.2P

RLOAD2 3 2102 50

CLOAD2 101 2102 0.2P

CLOAD3 2101 2102 0.05P

\*

RSHUNT 2101 2102 115

XPK1 2101 2102 8 9 0 0 0 INPKG

\*

XCIROUT 8 9 101 INBUF

\*

.SUBCKT INBUF 1001 1002 101

\*

\*

XQ1 101 1001 4 0 H11A03

XQ2 101 1002 5 0 H11A03

\*

R1 1 1001 400

R2 1 1002 400

\*R3 1 101 1200

XDS 101 1 0 HDU031011

\*

R3 4 0 7000

R4 5 0 7000

IB3 1 0 0.40M

\*

XPAD1 1001 101 0 0 HPAD3

XPAD2 1002 101 0 0 HPAD3

\*

.ENDS INBUF

\*

```

*
.SUBCKT INPKG 1101 1102 1201 1202 110 111 112
*
* resistors
*
R1201 1201 1203 260M
R1202 1202 1204 260M
*
* inductors
*
LLAP_1_3 1101 1203 1.02N
LLAP_2_4 1102 1204 1.15N
KLAP_1_2 LLAP_1_3 LLAP_2_4 0.17
*
* capacitors
*
C01 1101 0 145F
C02 1102 0 140F
*
* mutual capacitors
*
C01_02 1101 1102 43.800F
.ENDS INPKG
*
*
*BEGIN PROCESS MODELS
*
*
.SUBCKT H11A03 1 2 3 21
CP1EPI 1 2 2.448F
CP1P2 12 3 3.390F
CTRENCH 1 20 6.697F
RBX 2 12 125.448 TC=2.687M
RCX 1 10 81.681 TC=3.007M,1.816U
RCI 10 11 4.299 TC=3.007M,1.816U
REX 13 3 16.820 TC=69.810U
RSUB 20 21 16.621K
QP 20 10 12 20 TXP OFF
QN 11 12 13 11 TX
*XREPORT1 0 REPORTERL1N3
*XREPORT2 0 REPORTERL1N4
.MODEL TX NPN( IS=5.117E-018 XTI=3 EG=1.140 BF=246.294 BR=20 XTB=450M
+ VAF=29 VAR=3.500 NF=1.010 NR=1.020 NE=1.650 NC=1.560 IKF=19.257M
+ IKR=352.800U ISE=2.420E-021 ISC=3.112E-030 RB=125.448 RBM=94.086
+ IRB=3.112M CJE=16.685F MJE=490M VJE=940M FC=990M CJC=3.606F MJC=470M
+ VJC=850M TF=3.728P TR=19N XTF=1 VTF=1K ITF=9.305M PTF=5 KF=1.500F
+ AF=1 )
.MODEL TXP PNP( IS=3.300E-019 CJE=3.606F MJE=470M VJE=850M CJC=6.613F
+ MJC=400M VJC=650M BF=10K BR=593.542U TF=1N FC=900M )
.ENDS H11A03

.SUBCKT HDU031011 1 2 21
CM1EPI 1 4 2.260F
CTRENCH 2 5 5.461F
RS 4 2 29.945 TC=3.012M,2.350U

```

RSUB 5 21 20.159K  
DD 1 4 DSCH  
EVLEAK 1 10 1 4 990.871M  
DLEAK 4 10 DL  
DS 5 4 DSUB OFF  
\*XREPORT1 0 REPORTERL1N57  
\*XREPORT2 0 REPORTERL1N58  
.MODEL DSCH D( IS=161.314P N=1.095 CJO=21.747F M=370M VJ=680M EG=605M )  
.MODEL DSUB D( IS=4.044E-016 CJO=5.242F M=400M VJ=650M )  
.MODEL DL D( N=1 EG=400M IS=3.109N )  
.ENDS HDU031011

.SUBCKT HDE113032 1 2 21  
CP1EPI 1 4 88.881F  
QD 5 4 1 5 QESD  
RS 4 2 2.531 TC=2.729M,1.896U  
RSUB 5 21 2.936K  
CTRENCH 2 5 22.961F  
\*XREPORT1 0 REPORTERL1N63  
\*XREPORT2 0 REPORTERL1N64  
.MODEL QESD PNP( IS=1.181E-017 NF=1.050 BF=800M BR=600U CJE=155.018F  
+ VJE=600M MJE=400M CJC=53.258F VJC=650M MJC=400M )  
.ENDS HDE113032

.SUBCKT HPAD3 1 101 102 3  
CPAD 1 10 86.407F  
XQ1 1 101 3 HDE113032  
\* DE381011 is the ESD protection doide  
XQ2 102 1 3 HDE113032  
REPI 10 20 149.204M TC=4.800M,5U  
CTRENCH 21 20 79.795F  
DS 21 20 DSUB  
RS 3 21 369.115  
\*XREPORT1 0 REPORTERL1N91  
.MODEL DSUB D( IS=98.719F CJO=555.750F M=400M VJ=650M )  
.ENDS HPAD3

.PRINT TRAN V(8) V(9)  
\*.PRINT TRAN V(2101) V(2102)  
\*.PROBE  
.END